Rules for integrands of the form $(a + b\,\text{ArcSinh}[c\,x])^n$

1: $\int (a + b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n > 0$

- **Derivation: Integration by parts**
- **Rule: If $n > 0$, then**

$$\int (a + b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow x\,(a + b\,\text{ArcSinh}[c\,x])^n - b\,c\,n \int \frac{x\,(a + b\,\text{ArcSinh}[c\,x])^{n-1}}{\sqrt{1 + c^2\,x^2}}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_.,x_Symbol] :=
  x*(a+b*ArcSinh[c*x])^n -
  b*c*n*Int[x*(a+b*ArcSinh[c*x])^(n-1)/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c},x] && GtQ[n,0]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  x*(a+b*ArcCosh[c*x])^n -
  b*c*n*Int[x*(a+b*ArcCosh[c*x])^(n-1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c},x] && GtQ[n,0]
```

2: $\int (a + b\,\text{ArcSinh}[c\,x])^n\,dx$ when $n < -1$

- **Derivation: Integration by parts**
- **Basis:** $\frac{(a+b\,\text{ArcSinh}[c\,x])^n}{\sqrt{1+c^2\,x^2}} == \partial_x \frac{(a+b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n+1)}$
- **Rule: If $n < -1$, then**

$$\int (a + b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \frac{\sqrt{1 + c^2\,x^2}\,(a + b\,\text{ArcSinh}[c\,x])^{n+1}}{b\,c\,(n + 1)} - \frac{c}{b\,(n + 1)} \int \frac{x\,(a + b\,\text{ArcSinh}[c\,x])^{n+1}}{\sqrt{1 + c^2\,x^2}}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  Sqrt[1+c^2*x^2]*(a+b*ArcSinh[c*x])^(n+1)/(b*c*(n+1)) -
  c/(b*(n+1))*Int[x*(a+b*ArcSinh[c*x])^(n+1)/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c},x] && LtQ[n,-1]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  Sqrt[-1+c*x]*Sqrt[1+c*x]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  c/(b*(n+1))*Int[x*(a+b*ArcCosh[c*x])^(n+1)/(Sqrt[-1+c*x]*Sqrt[1+c*x]),x] /;
FreeQ[{a,b,c},x] && LtQ[n,-1]
```

3: $\int (a + b\,\text{ArcSinh}[c\,x])^n\,dx$

- **Derivation: Integration by substitution**
- **Basis:** $(a + b\,\text{ArcSinh}[c\,x])^n = \frac{1}{b\,c}\,(a + b\,\text{ArcSinh}[c\,x])^n\,\text{Cosh}\left[\frac{a}{b} - \frac{a+b\,\text{ArcSinh}[c\,x]}{b}\right]\,\partial_x\,(a + b\,\text{ArcSinh}[c\,x])$
- **Rule:**

$$\int (a + b\,\text{ArcSinh}[c\,x])^n\,dx \longrightarrow \frac{1}{b\,c}\,\text{Subst}\left[\int x^n\,\text{Cosh}\left[\frac{a}{b} - \frac{x}{b}\right]dx,\ x,\ a + b\,\text{ArcSinh}[c\,x]\right]$$

- **Program code:**

```
Int[(a_.+b_.*ArcSinh[c_.*x_])^n_,x_Symbol] :=
  1/(b*c)*Subst[Int[x^n*Cosh[a/b-x/b],x],x,a+b*ArcSinh[c*x]] /;
FreeQ[{a,b,c,n},x]
```

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  -1/(b*c)*Subst[Int[x^n*Sinh[a/b-x/b],x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,n},x]
```